ΕΠΙΛΥΣΗ ΚΑΙ ΜΕΛΕΤΗ ΔΙΑΦΟΡΙΚΩΝ ΕΞΙΣΩΣΕΩΝ ΜΕ MATHEMATICA

2024

1. Επαλήθευση λύσεων ΣΔΕ

1) Να δειχθεί ότι η συνάρτηση $\dfrac{\sin x}{x}$ είναι λύση της ΣΔΕ: $xy' + y = \cos x$, όπου $y = y(x)$.

2) Να δειχθεί ότι η συνάρτηση $\dfrac{\sin(3\ln x)}{x}$ είναι λύση της ΣΔΕ: $x^3y''' + x^2y'' + xy' - 40y = 0$, όπου $y = y(x)$.

3) Έστω ότι η συνάρτηση $y = y(x)$ ορίζεται πεπλεγμένα μέσω της σχέσης $(x^2 + y^2)^2 - 5xy = 0$.

α) Να δειχθεί ότι αυτή η συνάρτηση είναι λύση της ΣΔΕ

$$[4x(x^2 + y^2) - 5y]dx + [4y(x^2 + y^2) - 5x]dy = 0.$$

β) Να γίνει γραφική παράσταση αυτής της λύσης για $x, y \in [-2, 2]$.

2. Απευθείας επίλυση ΣΔΕ και διαχείριση των αποτελεσμάτων

1) Έστω η ΣΔΕ: $$xy'(x)+y(x)=x\sin x\,. \tag{1}$$

α) Να βρεθεί αναλυτικά και απευθείας η γενική λύση της (1)

β) Να επαληθευτεί ότι η συνάρτηση που βρήκατε στο προηγούμενο ερώτημα, επαληθεύει όντως την (1).

γ) Να γίνει, στο ίδιο σχήμα, γραφική παράσταση της λύσης της (1) για $x\in[1,3\pi]$ και για τιμές της αυθαίρετης σταθεράς ίσες με 1, -1 και 0 με αντίστοιχη χρωματική ένδειξη.

2) Έστω το ΠΑΤ: $$y'(x)-2y(x)=0,\ y(1)=a\,. \tag{2}$$

α) Να βρεθεί αναλυτικά και απευθείας η λύση του ΠΑΤ (2).

β) Να γίνει, στο ίδιο σχήμα, γραφική παράσταση της λύσης του (2) για $x\in[0,3]$ και για τιμές της αυθαίρετης σταθεράς από -2 έως 2 με βήμα 1 χωρίς να αντικατασταθούν οι τιμές της αυθαίρετης σταθεράς. Επιπλέον στο σχήμα να εμφανίζεται αντίστοιχη χρωματική ένδειξη με «λεζάντες» της μορφής “a = …”.

γ) Να λυθεί αριθμητικά το ΠΑΤ (2) για y(1) = -1, $x\in[0,3]$ και στη συνέχεια να γίνει γραφική παράσταση της λύσης που βρέθηκε.

δ) Στο ίδιο σχήμα να παρουσιασθούν η αριθμητική και η αναλυτική λύση του ΠΑΤ (2) για y(1) = -1, $x\in[0,3]$. Για την αναλυτική λύση να χρησιμοποιηθούν οι επιλογές Thick, Dotted και Red.

3) Έστω το ΠΣΤ: $$x^2y''(x)-12xy'(x)+42y(x)=0,\ y(1)=1,\ y(2)=-1 \tag{3}$$

α) Να βρεθεί αναλυτικά και απευθείας η λύση του ΠΣΤ (3)

β) Να επαληθευτεί ότι η συνάρτηση που βρήκατε στο προηγούμενο ερώτημα, επαληθεύει όντως τη ΣΔΕ του ΠΣΤ (3).

γ) Να γίνει η γραφική παράσταση της λύσης του (3) για $x\in[1,2]$.

4) Να βρεθεί αναλυτικά και απευθείας η λύση του ΠΑΤ:
$$y''(t)+y(t)=0,\ y(0)=0,\ y'(0)=1 \tag{4}$$
που περιγράφει το πρόβλημα ενός απλού αρμονικού ταλαντωτή. Στη συνέχεια να γίνει γραφική παράσταση της λύσης που βρήκατε για $t\in[0,6\pi]$.

5) Να λυθεί αριθμητικά το ΠΑΤ:
$$y''(t)+\sin y(t)=0,\ y(0)=0,\ y'(0)=1 \tag{5}$$
που περιγράφει το πρόβλημα ενός απλού εκκρεμούς. Στη συνέχεια να γίνει γραφική παράσταση της λύσης που βρήκατε για $t\in[0,6\pi]$ με κόκκινο χρώμα.

6) Να παρασταθούν στο ίδιο σχήμα, οι λύσεις των ΠΑΤ (4) και (5).

3. Απευθείας επίλυση ΣΣΔΕ και διαχείριση των αποτελεσμάτων

1) Να βρεθεί αναλυτικά και απευθείας η γενική λύση του ΣΣΔΕ:
$$\left.\begin{aligned} x'(t) &= x(t)+y(t) \\ y'(t) &= 4x(t)+y(t) \end{aligned}\right\} \tag{1}$$
και στη συνέχεια να γίνει επαλήθευση της λύσης.

2) Να βρεθεί αναλυτικά και απευθείας η λύση του ΣΣΔΕ (1) που ικανοποιεί τις συνθήκες $x(0)=1, y(0)=2$ και στη συνέχεια να γίνει στο ίδιο σχήμα (με σχετική χρωματική ένδειξη) η γραφική παράσταση της λύσης του για $t \in [0,2]$.

3) Να βρεθεί αναλυτικά και απευθείας η γενική λύση του ΣΣΔΕ:
$$\left.\begin{aligned} x'(t) &= -\frac{x(t)}{2}+y(t) \\ y'(t) &= -x(t)-\frac{y(t)}{2} \end{aligned}\right\}.$$

4) Να βρεθεί αναλυτικά και απευθείας η γενική λύση του ΣΣΔΕ:
$$\left.\begin{aligned} x'(t) &= x(t)-y(t) \\ y'(t) &= x(t)+3y(t) \end{aligned}\right\}.$$

5) Να βρεθεί αναλυτικά και απευθείας η γενική λύση του ΣΣΔΕ:
$$\left.\begin{aligned} x'(t) &= -3x(t)+2z(t) \\ y'(t) &= x(t)-y(t) \\ z'(t) &= -2\,x(t)-y(t) \end{aligned}\right\}.$$

6) Να λυθεί αριθμητικά το ΠΑΤ:
$$\left.\begin{aligned} x' &= y-x(x^2+y^2) \\ y' &= -x-y(x^2+y^2) \end{aligned}\right\}, x(0)=1, y(0)=0$$
όπου x = x(t), y = y(t). Στη συνέχεια, να γίνει στο ίδιο σχήμα η γραφική παράσταση της λύσης του για $t \in [0,100]$, καθώς και η γραφική παράσταση των τροχιών του στο πεδίο των φάσεων.

7) Να λυθεί αριθμητικά το ΠΑΤ:
$$\left.\begin{aligned} x' &= -10(x-y) \\ y' &= -xs+28x-y \\ s' &= xy-\frac{8s}{3} \end{aligned}\right\}, x(0)=0, y(0)=1, s(0)=0$$
όπου x = x(t), y = y(t), s = s(t) και στη συνέχεια να γίνει γραφική παράσταση των τροχιών του στα πεδίο των φάσεων xs, xy, ys, καθώς και xys.

ΕΠΑΛΗΘΕΥΣΗ ΛΥΣΕΩΝ

In[1]:=
```
ODE11 = x * y'[x] + y[x] == Cos[x]
```
συνημίτο

Out[1]= $y[x] + x\,y'[x] == \mathrm{Cos}[x]$

In[2]:=
```
y11[x_] = Sin[x] / x
```
ημίτονο

Out[2]= $\frac{\mathrm{Sin}[x]}{x}$

In[3]:=
```
ODE11 /. y → y11
```

Out[3]= $\frac{\mathrm{Sin}[x]}{x} + x\left(\frac{\mathrm{Cos}[x]}{x} - \frac{\mathrm{Sin}[x]}{x^2}\right) == \mathrm{Cos}[x]$

“Δεν ξέρουμε” ακόμη αν έγινε επαλήθευση.

In[4]:=
```
ODE11 /. y → y11 // FullSimplify
```
πλήρης απλοποίηση

Out[4]= True

Άρα η y11(x) είναι λύση της ODE1.

In[5]:=
```
ODE12 = x^3 * y'''[x] + x^2 * y''[x] + x * y'[x] - 40 * y[x] == 0
```

Out[5]= $-40\,y[x] + x\,y'[x] + x^2\,y''[x] + x^3\,y^{(3)}[x] == 0$

In[6]:=
```
y12[x_] = Sin[3 * Log[x]] / x
```
ημίτονο λογάριθμος

Out[6]= $\frac{\mathrm{Sin}[3\,\mathrm{Log}[x]]}{x}$

In[7]:=
```
ODE12 /. y → y12 // FullSimplify
```
πλήρης απλοποίηση

Out[7]= True

Άρα η y12(x) είναι λύση της ODE2.

In[8]:=
```
EQ11 = (x^2 + y^2)^2 - 5 * x * y == 0
```

Out[8]= $-5\,x\,y + \left(x^2 + y^2\right)^2 == 0$

In[10]:=
```
EQ12 = Dt[EQ11]
```
ολική παράγ

Out[10]= $-5\,y\,\mathrm{Dt}[x] - 5\,x\,\mathrm{Dt}[y] + 2\left(x^2 + y^2\right)\left(2\,x\,\mathrm{Dt}[x] + 2\,y\,\mathrm{Dt}[y]\right) == 0$

In[11]:=
```
Collect[EQ12, {Dt[x], Dt[y]}]
```
συλλογή εκφράσ··· ολική ··· ολική παρά

Out[11]= $\left(-5\,y + 4\,x\left(x^2 + y^2\right)\right)\mathrm{Dt}[x] + \left(-5\,x + 4\,y\left(x^2 + y^2\right)\right)\mathrm{Dt}[y] == 0$

In[12]:=
```
f11[x_, y_] = EQ11[[1]]
```

Out[12]= $-5\,x\,y + \left(x^2 + y^2\right)^2$

In[12]:=
```
ContourPlot[f11[x, y] == 0, {x, -2, 2}, {y, -2, 2}]
```
διάγραμμα ισοϋψών

Out[12]=



ΑΠΕΥΘΕΙΑΣ ΕΠΙΛΥΣΗ ΣΔΕ ΚΑΙ ΔΙΑΧΕΙΡΙΣΗ ΤΩΝ ΑΠΟΤΕΛΕΣΜΑΤΩΝ

In[13]:=
```
ODE21 = x * y'[x] + y[x] == x * Sin[x]
```
ημίτονο

Out[13]= $y[x] + x\,y'[x] == x\,\mathrm{Sin}[x]$

In[14]:=
```
sol21 = DSolve[ODE21, y[x], x]
```
λύση διαφορικής εξίσωσης

Out[14]= $\left\{\left\{y[x] \to \frac{\mathbb{C}_1}{x} + \frac{-x\,\mathrm{Cos}[x] + \mathrm{Sin}[x]}{x}\right\}\right\}$

In[15]:=
```
y21[x_] = sol21[[1, 1, 2]]
```

Out[15]= $\frac{\mathbb{C}_1}{x} + \frac{-x\,\mathrm{Cos}[x] + \mathrm{Sin}[x]}{x}$

In[16]:=
```
ODE21 /. y → y21 // FullSimplify
```
πλήρης απλοποίηση

Out[16]= True

```
In[17]:= Plot[{y21[x] /. C[1] → 1, y21[x] /. C[1] → -1, y21[x] /. C[1] → 0},
         διάγραμμα        προεπιλεγμένη μορφή ⋯  προεπιλεγμένη μορφή στ⋯  προεπιλεγμένι
          {x, 1, 3 * Pi}, PlotLegends → "Expressions"]
                   π      υπομνήματα διαγράμματος
```

Out[17]=




```
In[18]:= ODE22 = y'[x] - 2 * y[x] == 0
Out[18]= -2 y[x] + y′[x] == 0

In[19]:= sol22 = DSolve[{ODE22, y[1] == a}, y[x], x]
                 λύση διαφορικής εξίσωσης
```

Out[19]= $\{\{y[x] \to a\, e^{-2+2x}\}\}$

```
In[20]:= Plot[Evaluate[Table[sol22[[1, 1, 2]], {a, -2, 2}]], {x, 0, 3},
         διά⋯ αξιολόγησ⋯ πίνακας τιμών
          PlotLegends → {"a = -2", "a = -1", "a = 0", "a = 1", "a = 2"}]
          υπομνήματα διαγράμματος
```

Out[20]=




```
In[21]:= sol23 = NDSolve[{ODE22, y[1] == -1}, y, {x, 0, 3}]
                 αριθμητική λύση διαφορικής εξίσωσης

Out[21]= {{y → InterpolatingFunction[ Domain: {{0., 3.}} Output: scalar ]}}
```

In[22]:=
```
g21 = Plot[Evaluate[y[x] /. sol23], {x, 0, 3}, PlotRange → All]
```
διά⋯ αξιολόγηση ανεξαρτήτως περιορισμών εύρος διαγρ⋯ όλα

Out[22]=



In[23]:=
```
y22[x_] = sol22[[1, 1, 2]] /. a → -1
```

Out[23]= $-e^{-2+2x}$

In[24]:=
```
g22 = Plot[y22[x], {x, 0, 3}, PlotStyle → {Thick, Dotted, Red}]
```
διάγραμμα στυλ διαγράμμ⋯ παχύ διακεκ⋯ κόκκινο

Out[24]=



In[25]:=
```
Show[g21, g22]
```
προβολή γραφικών

Out[25]=



In[26]:=
```
ODE23 = x^2 * y''[x] - 12 * x * y'[x] + 42 * y[x] == 0
```

Out[26]= $42\, y[x] - 12\, x\, y'[x] + x^2\, y''[x] == 0$

In[27]:=
```
sol24 = DSolve[{ODE23, y[1] == 1, y[2] == -1}, y[x], x]
```
λύση διαφορικής εξίσωσης

Out[27]= $\left\{\left\{y[x] \to \frac{1}{64}\left(129\, x^6 - 65\, x^7\right)\right\}\right\}$

In[28]:=
```
y23[x_] = sol24[[1, 1, 2]]
```

Out[28]= $\frac{1}{64}\left(129\, x^6 - 65\, x^7\right)$

In[29]:=
```
ODE23 /. y → y23 // FullSimplify
```
πλήρης απλοποίηση

Out[29]= True

In[30]:=
```
Plot[y23[x], {x, 1, 2}]
```
διάγραμμα

Out[30]=


In[31]:=
```
ODE24 = y''[t] + y[t] == 0
```

Out[31]= $y[t] + y''[t] == 0$

In[32]:=
```
sol25 = DSolve[{ODE24, y[0] == 0, y'[0] == 1}, y[t], t]
```
λύση διαφορικής εξίσωσης

Out[32]= {{y[t] → Sin[t]}}

In[33]:=
```
g23 = Plot[sol25[[1, 1, 2]], {t, 0, 6 * Pi}]
```
διάγραμμα π

Out[33]=



In[34]:=
```
ODE25 = y''[t] + Sin[y[t]] == 0
```
ημίτονο

Out[34]=
```
Sin[y[t]] + y''[t] == 0
```

In[35]:=
```
sol26 = NDSolve[{ODE25, y[0] == 0, y'[0] == 1}, y, {t, 0, 6 * Pi}]
```
αριθμητική λύση διαφορικής εξίσωσης π

Out[35]=
```
{{y → InterpolatingFunction[ Domain: {{0., 18.8}} Output: scalar ]}}
```

In[36]:=
```
g24 = Plot[Evaluate[y[t] /. sol26], {t, 0, 6 * Pi}, PlotStyle → Red]
```
διά··· αξιολόγηση ανεξαρτήτως περιορισμών π στυλ διαγρά··· κόκκιν

Out[36]=



In[37]:= `Show[g23, g24]`
προβολή γραφικών

Out[37]= 

ΑΠΕΥΘΕΙΑΣ ΕΠΙΛΥΣΗ ΣΣΔΕ ΚΑΙ ΔΙΑΧΕΙΡΙΣΗ ΤΩΝ ΑΠΟΤΕΛΕΣΜΑΤΩΝ

In[38]:= `ODE31a = x'[t] == x[t] + y[t]`

Out[38]= $x'[t] == x[t] + y[t]$

In[39]:= `ODE31b = y'[t] == 4 * x[t] + y[t]`

Out[39]= $y'[t] == 4\, x[t] + y[t]$

In[40]:= `sol31 = DSolve[{ODE31a, ODE31b}, {x[t], y[t]}, t]`
λύση διαφορικής εξίσωσης

Out[40]= $\left\{\left\{x[t] \to \frac{1}{2} e^{-t} \left(1 + e^{4t}\right) c_1 + \frac{1}{4} e^{-t} \left(-1 + e^{4t}\right) c_2,\ y[t] \to e^{-t} \left(-1 + e^{4t}\right) c_1 + \frac{1}{2} e^{-t} \left(1 + e^{4t}\right) c_2\right\}\right\}$

In[41]:= `x31[t_] = sol31[[1, 1, 2]]`

Out[41]= $\frac{1}{2} e^{-t} \left(1 + e^{4t}\right) c_1 + \frac{1}{4} e^{-t} \left(-1 + e^{4t}\right) c_2$

In[42]:= `y31[t_] = sol31[[1, 2, 2]]`

Out[42]= $e^{-t} \left(-1 + e^{4t}\right) c_1 + \frac{1}{2} e^{-t} \left(1 + e^{4t}\right) c_2$

In[43]:= `ODE31a /. {x → x31, y → y31} // FullSimplify`
πλήρης απλοποίηση

Out[43]= True

In[44]:= `ODE31b /. {x → x31, y → y31} // FullSimplify`
πλήρης απλοποίηση

Out[44]= True

In[45]:= `sol32 = DSolve[{ODE31a, ODE31b, x[0] == 1, y[0] == 2}, {x[t], y[t]}, t]`
λύση διαφορικής εξίσωσης

Out[45]= $\left\{\left\{x[t] \to e^{3t},\ y[t] \to 2\, e^{3t}\right\}\right\}$

In[46]:= `x32[t_] = sol32[[1, 1, 2]]`

Out[46]= e^{3t}

In[47]:= `y32[t_] = sol32[[1, 2, 2]]`

Out[47]= $2\,e^{3t}$

In[48]:= `Plot[{x32[t], y32[t]}, {t, 0, 2}, PlotLegends → "Expressions"]`

διάγραμμα   υπομνήματα διαγράμματος

Out[48]=



In[49]:= `DSolve[{x'[t] == -x[t] / 2 + y[t], y'[t] == -x[t] - y[t] / 2}, {x[t], y[t]}, t]`

λύση διαφορικής εξίσωσης

Out[49]= $\{\{x[t] \to e^{-t/2}\, c_1 \,\mathrm{Cos}[t] + e^{-t/2}\, c_2\, \mathrm{Sin}[t],\ y[t] \to e^{-t/2}\, c_2\, \mathrm{Cos}[t] - e^{-t/2}\, c_1\, \mathrm{Sin}[t]\}\}$

In[50]:= `DSolve[{x'[t] == x[t] - y[t], y'[t] == x[t] + 3 * y[t]}, {x[t], y[t]}, t]`

λύση διαφορικής εξίσωσης

Out[50]= $\{\{x[t] \to -e^{2t}\,(-1 + t)\, c_1 - e^{2t}\, t\, c_2,\ y[t] \to e^{2t}\, t\, c_1 + e^{2t}\,(1 + t)\, c_2\}\}$

In[51]:=
```
DSolve[{x'[t] == -3*x[t] + 2*z[t], y'[t] == x[t] - y[t], z'[t] == -2*x[t] - y[t]},
 {x[t], y[t], z[t]}, t]
```
λύση διαφορικής εξίσωσης

Out[51]=

$$\Big\{\Big\{x[t] \to \frac{1}{3} e^{-2t} c_1 \left(2 + e^t \mathrm{Cos}\left[\sqrt{2}\, t\right] - 2\sqrt{2}\, e^t \mathrm{Sin}\left[\sqrt{2}\, t\right]\right) + \frac{1}{3} e^{-2t} c_2 \left(-2 + 2 e^t \mathrm{Cos}\left[\sqrt{2}\, t\right] - \sqrt{2}\, e^t \mathrm{Sin}\left[\sqrt{2}\, t\right]\right) + \frac{2}{3} e^{-2t} c_3 \left(-1 + e^t \mathrm{Cos}\left[\sqrt{2}\, t\right] + \sqrt{2}\, e^t \mathrm{Sin}\left[\sqrt{2}\, t\right]\right),$$

$$y[t] \to -\frac{1}{3} e^{-2t} c_3 \left(-2 + 2 e^t \mathrm{Cos}\left[\sqrt{2}\, t\right] - \sqrt{2}\, e^t \mathrm{Sin}\left[\sqrt{2}\, t\right]\right) + \frac{1}{3} e^{-2t} c_2 \left(2 + e^t \mathrm{Cos}\left[\sqrt{2}\, t\right] + \sqrt{2}\, e^t \mathrm{Sin}\left[\sqrt{2}\, t\right]\right) + \frac{1}{6} e^{-2t} c_1 \left(-4 + 4 e^t \mathrm{Cos}\left[\sqrt{2}\, t\right] + \sqrt{2}\, e^t \mathrm{Sin}\left[\sqrt{2}\, t\right]\right),$$

$$z[t] \to \frac{1}{3} e^{-2t} c_2 \left(-1 + e^t \mathrm{Cos}\left[\sqrt{2}\, t\right] - 2\sqrt{2}\, e^t \mathrm{Sin}\left[\sqrt{2}\, t\right]\right) + \frac{1}{3} e^{-2t} c_3 \left(-1 + 4 e^t \mathrm{Cos}\left[\sqrt{2}\, t\right] + \sqrt{2}\, e^t \mathrm{Sin}\left[\sqrt{2}\, t\right]\right) - \frac{1}{6} e^{-2t} c_1 \left(-2 + 2 e^t \mathrm{Cos}\left[\sqrt{2}\, t\right] + 5\sqrt{2}\, e^t \mathrm{Sin}\left[\sqrt{2}\, t\right]\right)\Big\}\Big\}$$

In[52]:=
```
DSolve[{x'[t] == -3*x[t] + 2*z[t], y'[t] == x[t] - y[t], z'[t] == -2*x[t] - y[t]},
 {x[t], y[t], z[t]}, t] // FullSimplify
```
λύση διαφορικής εξίσωσης

πλήρης απλοποίηση

Out[52]=

$$\Big\{\Big\{x[t] \to \frac{1}{3} e^{-2t} \left(2\,(c_1 - c_2 - c_3) + e^t \left(\left(c_1 + 2\,(c_2 + c_3)\right) \mathrm{Cos}\left[\sqrt{2}\, t\right] - \sqrt{2}\,\left(2 c_1 + c_2 - 2 c_3\right) \mathrm{Sin}\left[\sqrt{2}\, t\right]\right)\right),$$

$$y[t] \to \frac{1}{6} e^{-2t} \left(4\,(-c_1 + c_2 + c_3) + e^t \left(2\,\left(2 c_1 + c_2 - 2 c_3\right) \mathrm{Cos}\left[\sqrt{2}\, t\right] + \sqrt{2}\,\left(c_1 + 2\,(c_2 + c_3)\right) \mathrm{Sin}\left[\sqrt{2}\, t\right]\right)\right),\ z[t] \to \frac{1}{6} e^{-2t}$$

$$\left(2\,(c_1 - c_2 - c_3) + e^t \left(2\,(-c_1 + c_2 + 4 c_3) \mathrm{Cos}\left[\sqrt{2}\, t\right] + \sqrt{2}\,\left(-5 c_1 - 4 c_2 + 2 c_3\right) \mathrm{Sin}\left[\sqrt{2}\, t\right]\right)\right)\Big\}\Big\}$$

In[53]:=
```
sol33 = NDSolve[{x'[t] == y[t] - x[t] * (x[t]^2 + y[t]^2),
   y'[t] == -x[t] - y[t] * (x[t]^2 + y[t]^2), x[0] == 1, y[0] == 0}, {x[t], y[t]}, {t, 0, 100}]
```
αριθμητική λύση διαφορικής εξίσωσης

Out[53]=

{{x[t] → InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar][t],

y[t] → InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar][t]}}

In[54]:=
```
Plot[{Evaluate[x[t] /. sol33], Evaluate[y[t] /. sol33]}, {t, 0, 100}]
```
διάγ⋯ αξιολόγηση ανεξαρτήτως περιορ⋯ αξιολόγηση ανεξαρτήτως περιορισμών

Out[54]=


In[55]:=
```
ParametricPlot[{x[t], y[t]} /. sol33, {t, 0, 100}, AxesLabel → {x, y}]
```
παραμετρικό διάγραμμα ετικέτες αξόνων

Out[55]=


In[56]:=
```
sol34 = NDSolve[{x'[t] == -10 * (x[t] - y[t]),
   y'[t] == -x[t] * s[t] + 28 * x[t] - y[t], s'[t] == x[t] * y[t] - 8 * s[t] / 3,
   x[0] == 0, y[0] == 1, s[0] == 0}, {x[t], y[t], s[t]}, {t, 0, 100}]
```
αριθμητική λύση διαφορικής εξίσωσης

Out[56]=
```
{{x[t] → InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar ][t],
  y[t] → InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar ][t],
  s[t] → InterpolatingFunction[ Domain: {{0., 100.}} Output: scalar ][t]}}
```

In[57]:=
```
ParametricPlot[{x[t], s[t]} /. sol34, {t, 0, 100}, AxesLabel → {x, s}]
```
παραμετρικό διάγραμμα — ετικέτες αξόνων

Out[57]=



In[58]:=
```
ParametricPlot[{x[t], y[t]} /. sol34, {t, 0, 100}, AxesLabel → {x, y}]
```
παραμετρικό διάγραμμα   ετικέτες αξόνων

Out[58]=



In[59]:=
```
ParametricPlot[{y[t], s[t]} /. sol34, {t, 0, 100}, AxesLabel → {y, s}]
```
παραμετρικό διάγραμμα   ετικέτες αξόνων

Out[59]=



In[60]:=
```
ParametricPlot3D[{x[t], y[t], s[t]} /. sol34, {t, 0, 100}, AxesLabel → {x, y, s}]
```
3D παραμετρικό διάγραμμα — ετικέτες αξόνων

Out[60]=



4. Διαφορικές εξισώσεις με μερικές παραγώγους

Θεωρούμε τη μονοδιάστατη κυματική

$$\frac{\partial^2 u}{\partial t^2} = \frac{\partial^2 u}{\partial x^2} \qquad (1)$$

α) Να δειχθεί ότι η συνάρτηση $u(x,t) = \dfrac{\sin(x+t) - \sin(x-t)}{2} + \dfrac{e^{-(t-x)^2} - e^{-(t+x)^2}}{4}$ είναι λύση της (1).

β) Να γίνει γραφική παράσταση της u(x,t) για $x \in [-10,10]$, $t \in [0,6]$.

γ) Να γίνει στο ίδιο σχήμα γραφική παράσταση της u(x,t) για $x \in [-10,10]$ και t = 0, 1, 1.5, 2.5, 6 με αντίστοιχη χρωματική ένδειξη.

δ) Να γίνει «κινούμενη» γραφική παράσταση της u(x,t) για $x \in [-10,10]$, $t \in [0,12]$.

ΔΙΑΦΟΡΙΚΕΣ ΕΞΙΣΩΣΕΙΣ ΜΕ ΜΕΡΙΚΕΣ ΠΑΡΑΓΩΓΟΥΣ

In[1]:=
```
PDE11 = D[u[x, t], {t, 2}] == D[u[x, t], {x, 2}]
```
μερική παράγωγος | μερική παράγωγος

Out[1]= $u^{(0,2)}[x, t] == u^{(2,0)}[x, t]$

In[2]:=
```
u11[x_, t_] = (Sin[x + t] + Sin[x - t]) / 2 + (Exp[- (t - x) ^2] - Exp[- (t + x) ^2]) / 4
```
ημίτονο | ημίτονο | εκθετική συνάρτηση | εκθετική συνάρτηση

Out[2]= $\frac{1}{4}\left(e^{-(t-x)^2} - e^{-(t+x)^2}\right) + \frac{1}{2}\left(-\mathrm{Sin}[t-x] + \mathrm{Sin}[t+x]\right)$

In[3]:=
```
PDE11 /. u → u11 // FullSimplify
```
πλήρης απλοποίηση

Out[3]= True

In[4]:=
```
Plot3D[u11[x, t], {x, -2, 2}, {t, 0, 5}, AxesLabel → {x, y, u}]
```
3D διάγραμμα | ετικέτες αξόνων

Out[4]=


In[5]:=
```
Plot[{u11[x, 0], u11[x, 1], u11[x, 1.5], u11[x, 2.5], u11[x, 6]},
 {x, -10, 10}, PlotLegends → Automatic]
```
διάγραμμα · υπομνήματα δια··· · αυτόματο

Out[5]=



In[6]:=
```
Animate[Plot[u11[x, t], {x, -10, 10}, PlotRange → All], {t, 0, 12}]
```
εμψύχωση · διάγραμμα · εύρος διαγρ··· · όλα

Out[6]=



5. Μετασχηματισμός Laplace

1) Να βρεθούν
   α) Ο μετασχηματισμός Laplace μιας συνάρτησης f(t).
   β) Ο μετασχηματισμός Laplace της f(t), αλλά τώρα χρησιμοποιώντας το σύμβολο F(s) για το μετασχηματισμό Laplace.
   γ) Ο μετασχηματισμός Laplace της t^3.
   δ) Ο μετασχηματισμός Laplace της H(t-2).
   ε) Ο μετασχηματισμός Laplace των $\delta(t)$ και $\delta''(t)$.

2) Να βρεθεί ο αντίστροφος μετασχηματισμός Laplace των
   $\frac{1}{s-5}, \frac{1}{(s+1)^2}, \frac{1}{s^2+s+1}, \frac{1}{s(s+2)^2}, \frac{10s}{s^4+17s^2+16}, \frac{8se^{-3s}}{s^2+1}$

3) Γνωρίζει η Mathematica:
   α) τη γραμμική ιδιότητα του μετασχηματισμού Laplace;
   β) την ιδιότητα του μετασχηματισμού Laplace σύμφωνα με την οποία $L[f'''(t)] = s^3\,F(s) - s^2 f(0) - sf'(0) - f''(0)$;
   γ) την ιδιότητα του μετασχηματισμού Laplace σύμφωνα με την οποία $L[t^3 f(t)] = -F'''(s)$;

4) Να βρεθούν
   α) Ο μετασχηματισμός Laplace της συνάρτησης e^{x+2t} ως προς t.
   β) Ο μετασχηματισμός Laplace της ίδιας συνάρτησης ως προς x.
   γ) Ο μετασχηματισμός Laplace της συνάρτησης $u_t(x,t)$ ως προς t. Γνωρίζει η Mathematica τη σχετική ιδιότητα του μετασχηματισμού Laplace;
   δ) Ο μετασχηματισμός Laplace της συνάρτησης $u_x(x,t)$ ως προς t. Γνωρίζει η Mathematica τη σχετική ιδιότητα του μετασχηματισμού Laplace;

5) Με χρήση του μετασχηματισμού Laplace να λυθεί το ΠΑΤ
   $$y'''(t) - 4y''(t) - 9y'(t) + 36y(t) = 0,\ t > 0,\ y(0) = 1,\ y'(0) = 0,\ y''(0) = -1.$$

6) Με χρήση του μετασχηματισμού Laplace να λυθεί το πρόβλημα
   $$\frac{\partial c(x,t)}{\partial t} + V\frac{\partial c(x,t)}{\partial x} = 0,\ c(x,0) = 0,\ c(0,t) = c0$$

ΜΕΤΑΣΧΗΜΑΤΙΣΜΟΣ LAPLACE

In[1]:=
```
LaplaceTransform[f[t], t, s]
```
λαπλασιανός μετασχηματισμός

Out[1]= LaplaceTransform[f[t], t, s]

In[2]:=
```
LaplaceTransform[f[t], t, s] /. LaplaceTransform[f[t], t, s] → F[s]
```
λαπλασιανός μετασχηματισμός   λαπλασιανός μετασχηματισμός

Out[2]= F[s]

In[3]:=
```
LaplaceTransform[t^3, t, s]
```
λαπλασιανός μετασχηματισμός

Out[3]= $\frac{6}{s^4}$

In[4]:=
```
LaplaceTransform[UnitStep[t - 2], t, s]
```
λαπλασιανός μετασχ··· συνάρτηση βήματος

Out[4]= $\frac{e^{-2 s}}{s}$

In[5]:=
```
LaplaceTransform[DiracDelta[t], t, s]
```
λαπλασιανός μετασχ··· συνάρτηση Dirac δ

Out[5]= 1

In[6]:=
```
LaplaceTransform[D[DiracDelta[t], {t, 2}], t, s]
```
λαπλασιανός μετασχ··· ·· συνάρτηση Dirac δ

Out[6]= s^2

In[7]:= InverseLaplaceTransform$\left[\frac{1}{s-5}, s, t\right]$
αντίστροφος λαπλασιανός

Out[7]= $e^{5 t}$

In[8]:= InverseLaplaceTransform$\left[\frac{1}{(s+1)^2}, s, t\right]$
αντίστροφος λαπλασιανός

Out[8]= $e^{-t} t$

In[9]:= InverseLaplaceTransform$\left[\frac{1}{s^2+s+1}, s, t\right]$
αντίστροφος λαπλασιανός

Out[9]= $\frac{2 e^{-t/2} \text{Sin}\left[\frac{\sqrt{3} t}{2}\right]}{\sqrt{3}}$

In[10]:= InverseLaplaceTransform[$\frac{1}{s * (s + 2)^2}$, s, t]
αντίστροφος λαπλασιανός

Out[10]= $\frac{1}{4} e^{-2t} \left(-1 + e^{2t} - 2t\right)$

In[11]:= InverseLaplaceTransform[$\frac{10 * s}{s^4 + 17 * s^2 + 16}$, s, t]
αντίστροφος λαπλασιανός

Out[11]= $\frac{2}{3} (\text{Cos}[t] - \text{Cos}[4t])$

In[12]:= InverseLaplaceTransform[$\frac{8 * s * \text{Exp}[-3 * s]}{s^2 + 1}$, s, t]
αντίστροφος λαπλασιανός

Out[12]= 8 Cos[3 - t] HeavisideTheta[-3 + t]

In[13]:= LaplaceTransform[a * f[t] + b * g[t], t, s]
λαπλασιανός μετασχηματισμός

Out[13]= a LaplaceTransform[f[t], t, s] + b LaplaceTransform[g[t], t, s]

H Mathematica γνωρίζει τη γραμμική ιδιότητα του μετασχηματισμού Laplace.

In[14]:= LaplaceTransform[f'''[t], t, s] /. LaplaceTransform[f[t], t, s] → F[s]
λαπλασιανός μετασχηματισμός λαπλασιανός μετασχηματισμός

Out[14]= $-s^2 f[0] + s^3 F[s] - s f'[0] - f''[0]$

H Mathematica γνωρίζει τη σχετική ιδιότητα του μετασχηματισμού Laplace.

In[15]:= LaplaceTransform[t^3 * f[t], t, s]
λαπλασιανός μετασχηματισμός

Out[15]= LaplaceTransform[t^3 f[t], t, s]

H Mathematica ΔΕΝ γνωρίζει τη σχετική ιδιότητα του μετασχηματισμού Laplace.

In[16]:= LaplaceTransform[Exp[x + 2 * t], t, s]
λαπλασιανός μετασχ··· εκθετική συνάρτηση

Out[16]= $\frac{e^x}{-2 + s}$

In[17]:= LaplaceTransform[Exp[x + 2 * t], x, s]
λαπλασιανός μετασχ··· εκθετική συνάρτηση

Out[17]= $\frac{e^{2t}}{-1 + s}$

In[18]:= LaplaceTransform[D[u[x, t], {t, 1}], t, s]
λαπλασιανός μετασχ··· μερική παράγωγος

Out[18]= s LaplaceTransform[u[x, t], t, s] - u[x, 0]

H Mathematica γνωρίζει τη σχετική ιδιότητα του μετασχηματισμού Laplace.

In[19]:= `LaplaceTransform[D[u[x, t], {x, 1}], t, s]`
λαπλασιανός μετασχ··· μερική παράγωγος

Out[19]= $\text{LaplaceTransform}\left[u^{(1,0)}[x,t],t,s\right]$

Η Mathematica ΔΕΝ γνωρίζει τη σχετική ιδιότητα του μετασχηματισμού Laplace.

In[20]:= `ODE61 = y'''[t] - 4 * y''[t] - 9 * y'[t] + 36 * y[t] == 0`

Out[20]= $36\,y[t]-9\,y'[t]-4\,y''[t]+y^{(3)}[t]==0$

In[21]:= `IC61 = {y[0] → 1, y'[0] → 0, y''[0] → -1}`

Out[21]= $\{y[0]\to 1,\ y'[0]\to 0,\ y''[0]\to -1\}$

In[22]:= `EQ61 = LaplaceTransform[ODE61, t, s]`
λαπλασιανός μετασχηματισμός

Out[22]= $$36\,\text{LaplaceTransform}[y[t],t,s]+s^3\,\text{LaplaceTransform}[y[t],t,s]-9\left(s\,\text{LaplaceTransform}[y[t],t,s]-y[0]\right)-s^2\,y[0]-4\left(s^2\,\text{LaplaceTransform}[y[t],t,s]-s\,y[0]-y'[0]\right)-s\,y'[0]-y''[0]==0$$

In[23]:= `EQ61 = LaplaceTransform[ODE61, t, s] /. LaplaceTransform[y[t], t, s] → Y[s]`
λαπλασιανός μετασχηματισμός λαπλασιανός μετασχηματισμός

Out[23]= $$-s^2\,y[0]+36\,Y[s]+s^3\,Y[s]-9\left(-y[0]+s\,Y[s]\right)-4\left(-s\,y[0]+s^2\,Y[s]-y'[0]\right)-s\,y'[0]-y''[0]==0$$

In[24]:= `EQ62 = EQ61 /. IC61`

Out[24]= $1-s^2+36\,Y[s]+s^3\,Y[s]-9\left(-1+s\,Y[s]\right)-4\left(-s+s^2\,Y[s]\right)==0$

In[25]:= `EQ63 = EQ62 // Expand`
επέκτασr

Out[25]= $10+4\,s-s^2+36\,Y[s]-9\,s\,Y[s]-4\,s^2\,Y[s]+s^3\,Y[s]==0$

In[26]:= `sol61 = Solve[EQ63, Y[s]]`
λύση εξισώσεων και ανισ

Out[26]= $$\left\{\left\{Y[s]\to\frac{-10-4\,s+s^2}{36-9\,s-4\,s^2+s^3}\right\}\right\}$$

In[27]:= `y61[t_] = InverseLaplaceTransform[sol61[[1, 1, 2]], s, t]`
αντίστροφος λαπλασιανός

Out[27]= $$-\frac{1}{42}\,e^{-3\,t}\left(-11-91\,e^{6\,t}+60\,e^{7\,t}\right)$$

Ανάλυση σε απλά κλάσματα γίνεται με την εντολή Apart

In[28]:= `Apart[sol61[[1, 1, 2]]]`
αποσυναρμολόγηση

Out[28]= $$-\frac{10}{7\,(-4+s)}+\frac{13}{6\left(-3+s\right)}+\frac{11}{42\left(3+s\right)}$$

In[29]:= `PDE61 = D[c[x, t], t] + V * D[c[x, t], x] == 0`
μερική παράγωγος μερική παράγωγος

Out[29]= $c^{(0,1)}[x,t]+V\,c^{(1,0)}[x,t]==0$

In[30]:=
```
IC62 = {c[x, 0] → 0, LaplaceTransform[c[x, t], t, s] → C[x, s]}
```
λαπλασιανός μετασχηματισμός προεπιλεγμ

Out[30]= {c[x, 0] → 0, LaplaceTransform[c[x, t], t, s] → C[x, s]}

In[31]:=
```
EQ64 = LaplaceTransform[PDE61, t, s] /. IC62
```
λαπλασιανός μετασχηματισμός

Out[31]= $s\,C[x, s] + V\,\text{LaplaceTransform}\left[c^{(1,0)}[x, t], t, s\right] == 0$

In[32]:=
```
EQ65 = EQ64 /. LaplaceTransform[c^(1,0)[x, t], t, s] → D[C[x, s], x]
```
λαπλασιανός μετασχηματισμός ·· προεπιλεγμένη

Out[32]= $s\,C[x, s] + V\,C^{(1,0)}[x, s] == 0$

In[33]:=
```
Con62 = c[0, t] == c0
```

Out[33]= c[0, t] == c0

In[34]:=
```
EQ66 = LaplaceTransform[Con62, t, s]
```
λαπλασιανός μετασχηματισμός

Out[34]= $\text{LaplaceTransform}[c[0, t], t, s] == \frac{c0}{s}$

In[35]:=
```
EQ67 = LaplaceTransform[Con62, t, s] /. LaplaceTransform[c[0, t], t, s] → C[0, s]
```
λαπλασιανός μετασχηματισμός λαπλασιανός μετασχηματισμός προεπιλεγ

Out[35]= $C[0, s] == \frac{c0}{s}$

In[36]:=
```
sol62 = DSolve[{EQ65, EQ67}, C[x, s], x]
```
λύση διαφορικής εξίσωσης προεπιλεγμένη

Out[36]= $\left\{\left\{C[x, s] \to \frac{c0\, e^{-\frac{s\,x}{V}}}{s}\right\}\right\}$

In[37]:=
```
InverseLaplaceTransform[sol62[[1, 1, 2]], s, t]
```
αντίστροφος λαπλασιανός

Out[37]= $c0\,\text{HeavisideTheta}\left[t - \frac{x}{V}\right]$

6. Μετασχηματισμός Fourier

1) Να βρεθούν

   α) Ο μετασχηματισμός Fourier μιας συνάρτησης f(x).

   β) Ο μετασχηματισμός Fourier της f(x), αλλά τώρα χρησιμοποιώντας το σύμβολο F(ω) για το μετασχηματισμό Fourier.

   γ) Ο μετασχηματισμός Fourier της $e^{-|x|}$.

   δ) Ο μετασχηματισμός Fourier της $\delta(x)$.

2) Να βρεθεί ο αντίστροφος μετασχηματισμός Fourier των

   $e^{-|\omega|}, e^{-4\omega^2}, \frac{1}{1+i\omega}$

3) Γνωρίζει η Mathematica:

   α) την ιδιότητα του μετασχηματισμού Fourier σύμφωνα με την οποία $F[f(x-a)] = e^{ia\omega}F(\omega)$;

   β) τη γραμμική ιδιότητα του μετασχηματισμού Fourier;

   γ) την ιδιότητα του μετασχηματισμού Fourier σύμφωνα με την οποία $F[f''(x)] = -\omega^2 F(\omega)$;

4) Να βρεθούν

   α) Ο μετασχηματισμός Fourier της συνάρτησης της συνάρτησης $u_x(x,t)$ ως προς x. Γνωρίζει η Mathematica τη σχετική ιδιότητα του μετασχηματισμού Fourier;

   β) Ο μετασχηματισμός Fourier της συνάρτησης $u_t(x,t)$ ως προς x. Γνωρίζει η Mathematica τη σχετική ιδιότητα του μετασχηματισμού Fourier;

5) Με χρήση του μετασχηματισμού Fourier ως προς x να λυθεί το πρόβλημα

$$\frac{\partial s(x,t)}{\partial t} = \frac{\partial^2 s(x,t)}{\partial x^2}, s(x,0) = e^{-x^2},\ x \in \mathbb{R}.$$

ΜΕΤΑΣΧΗΜΑΤΙΣΜΟΣ FOURIER

In[1]:=
```
FourierTransform[f[x], x, ω]
```
μετασχηματισμός Fourier

Out[1]= FourierTransform[f[x], x, ω]

In[2]:=
```
FourierTransform[f[x], x, ω] /. FourierTransform[f[x], x, ω] → F[ω]
```
μετασχηματισμός Fourier μετασχηματισμός Fourier

Out[2]= F[ω]

In[3]:=
```
FourierTransform[Exp[-Abs[x]], x, ω]
```
μετασχηματισμός Fou·· εκθε··· απόλυτη τιμή

Out[3]= $\dfrac{\sqrt{\frac{2}{\pi}}}{1+\omega^2}$

Το αποτέλεσμα διαφέρει από το αντίστοιχο του τυπολογίου μας. Αυτό οφείλεται στον διαφορετικό ορισμό που χρησιμοποιεί η *Mathematica* για τον υπολογισμό του μετασχηματισμού Fourier. Για να έχουμε τα ίδια αποτελέσματα με το τυπολόγιό μάς πρέπει να το δηλώσουμε μέσω της εντολής:

In[4]:=
```
FourierTransform[Exp[-Abs[x]], x, ω, FourierParameters -> {1, -1}]
```
μετασχηματισμός Fou·· εκθε··· απόλυτη τιμή επιλογή παραμέτρων Fourier

Out[4]= $\dfrac{2}{1+\omega^2}$

ή εναλλακτικά

In[5]:=
```
SetOptions[{FourierTransform, InverseFourierTransform},
  FourierParameters → {1, -1}];
```
ορισμός επιλ··· μετασχηματισμός Four·· αντίστροφος μετασχηματισμός σειράς

επιλογή παραμέτρων Fourier

In[6]:=
```
FourierTransform[Exp[-Abs[x]], x, ω]
```
μετασχηματισμός Fou·· εκθε··· απόλυτη τιμή

Out[6]= $\dfrac{2}{1+\omega^2}$

In[7]:=
```
FourierTransform[DiracDelta[x], x, ω]
```
μετασχηματισμός Fou·· συνάρτηση Dirac δ

Out[7]= 1

In[8]:=
```
InverseFourierTransform[Exp[-Abs[ω]], ω, x]
```
αντίστροφος μετασχηματισμός σ·· εκθε··· απόλυτη τιμή

Out[8]= $\dfrac{1}{\pi\left(1+x^2\right)}$

```
In[9]:= InverseFourierTransform[Exp[-4 * ω^2], ω, x]
        αντίστροφος μετασχηματισμός σ··· εκθετική συνάρτηση
```

Out[9]= $\frac{e^{-\frac{x^2}{16}}}{4\sqrt{\pi}}$

```
In[10]:= InverseFourierTransform[1/(1 + i * ω), ω, x]
         αντίστροφος μετασχηματισμός σειράς Fourier
```

Out[10]= e^{-x} `HeavisideTheta[x]`

```
In[11]:= FourierTransform[f[x - a], x, ω]
         μετασχηματισμός Fourier
```

```
Out[11]= FourierTransform[f[-a + x], x, ω]
```

Η Mathematica ΔΕΝ γνωρίζει τη σχετική ιδιότητα του μετασχηματισμού Fourier.

```
In[12]:= FourierTransform[a * f[x] + b * g[x], x, ω]
         μετασχηματισμός Fourier
```

```
Out[12]= FourierTransform[a f[x] + b g[x], x, ω]
```

Η Mathematica ΔΕΝ γνωρίζει τη γραμμική ιδιότητα του μετασχηματισμού Fourier.

```
In[13]:= FourierTransform[f''[x], x, ω] /. FourierTransform[f[x], x, ω] → F[ω]
         μετασχηματισμός Fourier                μετασχηματισμός Fourier
```

Out[13]= $-\omega^2\, F[\omega]$

```
In[14]:= FourierTransform[D[u[x, t], {x, 1}], x, ω]
         μετασχηματισμός Fou··· μερική παράγωγος
```

```
Out[14]= i ω FourierTransform[u[x, t], x, ω]
```

Η Mathematica γνωρίζει τη σχετική ιδιότητα του μετασχηματισμού Fourier.

```
In[15]:= FourierTransform[D[u[x, t], {t, 1}], x, ω]
         μετασχηματισμός Fou··· μερική παράγωγος
```

Out[15]= `FourierTransform[`$u^{(0,1)}$`[x, t], x, ω]`

Η Mathematica ΔΕΝ γνωρίζει τη σχετική ιδιότητα του μετασχηματισμού Fourier.

```
In[16]:= PDE71 = D[s[x, t], t] == D[s[x, t], {x, 2}]
                 μερική παράγωγος   μερική παράγωγος
```

Out[16]= $s^{(0,1)}[x, t] == s^{(2,0)}[x, t]$

```
In[17]:= ICF = {FourierTransform[s[x, t], x, ω] → S[t, ω]}
               μετασχηματισμός Fourier
```

```
Out[17]= {FourierTransform[s[x, t], x, ω] → S[t, ω]}
```

```
In[18]:= EQ71 = FourierTransform[PDE71[[1]], x, ω] == FourierTransform[PDE71[[2]], x, ω] /. ICF
                μετασχηματισμός Fourier                 μετασχηματισμός Fourier
```

Out[18]= `FourierTransform[`$s^{(0,1)}$`[x, t], x, ω]` $== -\omega^2\, S[t, \omega]$

```
In[19]:= EQ72 = EQ71 /. FourierTransform[s^(0,1)[x, t], x, ω] → D[S[t, ω], t]
                        μετασχηματισμός Fourier                 μερική παράγωγος
```

Out[19]= $S^{(1,0)}[t, \omega] == -\omega^2 S[t, \omega]$

```
In[20]:= Con2 = s[x, 0] == Exp[-x^2]
                           εκθετική συνάρ
```

Out[20]= $s[x, 0] == e^{-x^2}$

```
In[21]:= EQ73 = S[0, ω] == FourierTransform[Con2[[2]], x, ω]
                           μετασχηματισμός Fourier
```

Out[21]= $S[0, \omega] == e^{-\frac{\omega^2}{4}} \sqrt{\pi}$

```
In[22]:= sol71 = DSolve[{EQ72, EQ73}, S[t, ω], t]
                 λύση διαφορικής εξίσωσης
```

Out[22]= $\left\{\left\{S[t, \omega] \to e^{-\frac{\omega^2}{4} - t\,\omega^2} \sqrt{\pi}\right\}\right\}$

```
In[23]:= s71[x_, t_] = InverseFourierTransform[sol71[[1, 1, 2]], ω, x] // FullSimplify
                       αντίστροφος μετασχηματισμός σειράς Fourier            πλήρης απλοποίη
```

Out[23]= $\dfrac{e^{-\frac{x^2}{1+4t}}}{\sqrt{1+4t}}$